```
<プログラム>::= <変数宣言部> <関数群> | <関数群>
```

<変数宣言部>::= <宣言文> <変数宣言部> | <宣言文>

<宣言文> ::= define <識別子>; | define <宣言配列>;

<宣言配列>::= <識別子><定数添字列>

<定数添字列>::=[数]|[数]<定数添字列>

<関数群>::= <関数><関数群> | <関数>

<文集合>::= <文> <文集合> | <文>

<文>::= <代入文> | <ループ文> | <条件分岐文> | <関数文>

<関数文>::= <識別子>(<引数列>); | <識別子>(); /* 文集合から呼び出し */

<関数>::= <識別子>(<仮引数列>){<変数宣言部><文集合>} | <識別子>(){<変数宣言部><文集合>} 合>}

<仮引数列>::=<仮引数><仮引数列>

<仮引数>::=<識別子> | <識別子>[]

<引数列>::= <引数>, <引数列> | <引数>

<引数>::= <算術式>

<代入文>::= <識別子> = <算術式>; | <識別子><増減演算子>; | <配列> = <算術式>; | <配列><増減演算子>

<代入式>::= <識別子> = <算術式> | <増減変数> | <配列> = <算術式>

<算術式>::= <算術式> <加減演算子> <項> | <項>

<項>::= <項> <乗除演算子> <因子> | <因子>

<因子>::= <変数> | (<算術式>) | <配列>

<加減演算子> ::= + | -

<乗除演算子>::=*|/

<増減演算子> ::= ++ | --

<変数>::= <識別子> | <数>

<増減変数> ::= <識別子><増減演算子> | <増減演算子><識別子> | <配列><増減演算子> | <増減演算子><配列>

<ループ文> ::= while(<条件式>){<文集合>} | for(<識別子> = <数>;<条件式>;<代入式>){< 文集合>)}| for(<配列> = <数>;<条件式>;<代入式>){<文集合>)}

<配列>::= <識別子><添字列>

<添字列>::=[<変数>]|[<変数>]<添字列>|[<配列>]<添字列>

<条件分岐文> ::= if(<条件式>){<文集合>} | if(<条件式>){<文集合>}else{<文集合>}

<条件式>::= <算術式><比較演算子><算術式>

<比較演算子> ::= == | '<' | '>' | != | '<'= | '>'=

<識別子>::= <英字> <英数字列> | <英字>

<英数字列>::= <英数字> <英数字列> | <英数字>

<英数字>::= <英字> | <数>

<数>::= <自然数字><準数> | <数字>

<準数>::= <数字><準数> | <数字>

<自然数字>::= 1|2|3|4|5|6|7|8|9

<英字>::= a|b|c|d|e|f|g|h|i|j|k|l|m|n|o|p|q|r|s|t|u|v|w|x|y|z|

A|B|C|D|E|F|G|H|I|J|K|L|M|N|O|P|Q|R|S|T|U|V|W|X|Y|Z

<数字>::= 0|<自然数字>